Ship roll motion model

Course:	IP500515 Modelling and Simulation of Dynamic Systems
Date:	09/12/2020
Name:	Ravitej Bhagavathi
Studno:	546691
Project:	Project No: 8 Ship Roll Motion Model

1. **Systematic modelling approach:**

1.1 System identification and objectives for modelling:

Roll motion analysis of a ship is critical for understanding its crew habitability, stability and manoeuvrability. The main objective for this modelling project is to simulate the roll motion of a ship in various sea conditions. For this purpose, the ship hull is identified as the "system" and the intact ship hull forms the border for the system.

1.2 Interaction with the system surroundings:

The surrounding water in which the ship floats, exerts hydrostatic and hydro-dynamic forces on the ship hull. The forces and the corresponding moments responsible for the ship's dynamic behaviour are shown in Fig.1



Fig.1 Various force acting on the ship structure

1.3 Assumptions and simplifications: The following minor effects are identified and omitted during the system modelling

- The ship has zero forward speed
- Wave amplitudes, if any, are small compared to the wave length. Wave slope is small.
- Wave length is much larger compared to the ship width
- The are no coupling effects between the ship's degrees of freedom.
- Diffraction forces which are caused because of the surrounding fluid acceleration are neglected and only the Froude-Krylov forces which are caused by the changing free surface is considered.
- Wave induced radiation damping is considered and all other damping effects like friction, eddy-making, bilge-keel and lift are neglected.
- Non-linear damping and restoring forces are neglected as the roll amplitudes are small
- Added mass moment of inertia is considered to be 20% of the total moment of inertia.
- Transverse metacentric height is assumed not to change with roll angle

1.4 State variables:

As the model is prepared to simulate the ship's roll motion, the roll angle (φ) and angular momentum (L) about the ship's longitudinal axis will be treated as system state variables.

1.5 Equations governing roll motion: The following equation is governing equation of roll motion with all the terms explained below [1]

$$I'_{xx}\,\ddot{\varphi} + b\,\dot{\varphi} + \Delta\,GM_T\,\varphi = \Delta\,GM_T\,\alpha_M\sin(\omega t)$$

I'_{xx} is the mass moment of the vessel about it's longitudinal axis

b is the radiation damping coefficient (linear only)

Δ is the vessel displacement

GM_T is the initial metacentric height

$\Delta\ GM_T$ is the stiffness coefficient due to the restoring force

$\Delta\ GM_T\ \alpha_M\ \sin(\omega t)$ is the wave excitation moment

2. **Basic elements:**

The basic elements of the system are the ship hull which represents the inertial element, the surrounding water which contributes to the virtual inertia, resistance and capacitance effects. The model is similar to a spring-mass-damper system.

3. **Bond graph model for each component:**

Component	B.G element	Equation
Ship hull + added inertia		$M = I_{xx}\ddot{\varphi}$ $e = I.p$
Damping moment		$M = b\dot{\varphi}$ $e = R.f$
Restoring moment		$M = \Delta GM\varphi$ $e = q/c$ $(c = 1/k)$
Excitation moment		$M = \Delta\ GM_T\ \alpha_M\ \sin(\omega t)$

4. **Ship parameters:**

For the purpose of this simulation, publicly available data of "S175 Container Ship" is chosen. S175 is a standard model recommended by ITTC (International towing tank committee) for the validation and comparison of various modelling experiments. The main particulars and parameters required for the current simulation are listed below:

L.B.P	175 m
Beam (amidships)	25.4 m
Draught (amidships)	9.50 m
Waterplane area	3157 m^2
Displacement	24743.5 m^3
Water density	1025 kg/m^3
LCG (aft midship)	2.34 m
VCG (above baseline)	9.52 m
Radius of Inertia, roll	8.30 m
Block Coefficient	0.57
GM_T	1.00 m

5. **Bond graph model in 20-Sim for free roll decay test:**

Using the basic elements and individual sub-models described in section 2 and section 3, the full-model is assembled in 20-Sim software[2] as shown in Fig.2



Fig.2 Bond graph model of ship in free roll decay test

Parameters, variables and equations for the individual components are assigned using the ship parameters and roll equations described in section 3 and section 1.5 respectively. The same are entered in the below 20-Sim equations editor.

Excitation Force: Since the current section is simulating a free roll decay in calm water, the excitation moment is zero

```
parameters
   real effort = 0;
variables
   real flow;
equations
   p.e = effort;
   flow = p.f;
```

Ship hull mass moment of inertia: The moment of inertia is calculated by using the hull parameters and assuming the added inertia is 20% of the actual inertia [1].

```
parameters
   real global Displacement = 24743.5;
   real global k_xx = 8.3;

variables
   real global I_xx;
   real i;

equations
   I_xx = 1.2*Displacement*k_xx^2;
   i = I_xx;
   state = int(p.e);
   p.f = state / i;
```

Radiation damping: The ship in roll will create waves in the surrounding water and ultimately lose its initial energy. For simplicity, the damping moment is assumed to be linear though it is not the case in actual scenario. The damping coefficient is derived from the damping ratio and critical damping coefficient values. From literature it is found that the damping ratio for Containers is in the range of 2-10% [3]. A value of 2% is used for the present simulation.

Damping ratio $\zeta = 0.02$
Critical damping $B_{cr} = 2\omega I$

Actual damping coefficient $b = B_{cr}\zeta$

```
parameters
    real neta = 0.02;
    real global Displacement;
    real global GM;

variables
    real global I_xx;
    real w_n;
    real r;

equations
    w_n = sqrt(Displacement*GM/I_xx);
    r = neta*2*w_n*I_xx;
    p.e = r * p.f;
```

Restoring moment: When a ship rolls about it's CG, the centre of buoyancy shifts towards the heeling side and this creates a restoring moment about the ship's CG. The equation for restoring moment is described in section 3.

```
parameters
    real global Displacement;
    real global density = 1025;
    real global GM = 1;

variables
    real c;
    real K;
    real deg;

equations
    K = Displacement*GM;
    c = 1/K;
    state = int(p.f);
    p.e = state / c;
    deg = state*180/pi;
```

Initial conditions: For the free roll decay test, initial energy is supplied in the way of a roll angle. In the current simulation, an initial roll angle of 5 degrees is chosen arbitrarily. The is input as the initial state in 20-Sim parameters editor.

$\varphi(0) = 5deg$

Simulation time: 2000 seconds to visualize the entire roll decay.

Solver routine: Backward Differentiation Formula

Results: The results of the simulation are shown in Fig. 3



Fig. 3 Free roll decay simulation plot

6. **Turing the parameters:**

The results of the previous simulation indicate that the vessel takes approximately 1500 sec to reach a steady state. This is not a "good" roll response. We can improve the vessel behaviour by tuning certain parameters like the damping ratio(ζ) and initial metacentric height (GM)

Damping ratio $\zeta = 0.1$ (upper limit for container vessels)
GM = 2

Results : The improved roll response of the vessel can be seen from Fig. 4



Fig.4 Roll response of the tuned model

Optional tasks

7. **Regular beam seas:**

The roll motion of the ship is analysed by applying a regular wave field. The excitation moment from the wave field is given by the relation

$M = \Delta\, GM_T\, \alpha_M \sin(\omega t) \longrightarrow \quad (1)$

$\alpha_M = \dfrac{2\pi\zeta_a}{L_w}$

ζ_a = Wave amplitude

L_w = Wave length

ω = wave frequency

A bond graph model is assembled in 20-Sim as shown below in Fig.5. It is similar to the model in section 6 with modified damping coefficient and effort source.



Fig.5 Bond graph model for the ship in regular wave field

Restoring moment and Inertia components remain the same as given in section 5. However, the damping coefficient and exciting moments are changed to reflect the wave loading.

Damping coefficient: Ikeda et.al developed empirical relations to predict the various damping coefficients for a ship in a random seaway.

$$B_{44} = B_{44}^F + B_{44}^W \dot{\eta} + B_{44}^E + B_{44}^{BK} + B_{44}^L$$

In the current simulation, only the second term - radiation wave damping (B_{44}^W) is considered, as it cause the most signifant damping effect for a ship with zero speed. The radiation wave damping relation is given below [ref]

$$\frac{b_{44}}{\rho A B^2}\sqrt{\frac{B}{2g}} = a(B/T)\exp\left(b(B/T)\varpi^{-1.3}\right)\varpi^{d(B/T)}$$

The MR component in the Bond Graph model is updated considering the above relation. The input wave frequency from the wave generator is passed on as an input to the MR component.

```
parameters
   real B = 25.4;
   real T = 9.5;
   real L = 175;
   real global g = 9.81;
   real global density;


variables
   real r;
   real b_44;
   real A;


equations

   A = B*T;
   b_44 = sqrt(2*g/B)*density*A*B^2*(0.256*(B/T)-0.286)*exp((-0.11*(B/T)-2.55)*(w^(-1.3)))*(w^(0.33*(B/T)-0.459));
   r = L*b_44;
   p.e = r * p.f;
```

Wave generator: A regular wave of given amplitude and time period is created using a regular sinusoidal wave equation. The outputs of the wave generator are wave frequency (ω) and wave signal. They are passed to the damping and exciation components of the model

```
parameters
   real amplitude = 5;        // amplitude of the wave
   real T = 20;      // Time period of the wave

variables
   real omega;


equations
   omega = 2*pi/T;
   "calculate the sine wave"
   wave[1,1] = amplitude * sin ( omega * time);
   wave[2,1] = omega;

   w = omega;
```

Excitation moment: The input wave is used to calculate the moment due to waves as given in eq. (1)

```
parameters
   real global Displacement;
   real global GM;
   real global g;

variables
   real flow;
   real omega;
   real wave_load;
   real wave;


equations
   omega = input[2,1];
   wave = input[1,1];
   wave_load = Displacement*GM*omega^2*wave/g;
   p.e = wave_load;
   flow = p.f;
```

For the above parameters, the simulation is performed at different wave periods and amplitudes and the results are plotted as shown Fig.6 to Fig. 10



Fig. 6 Roll response of a ship in a regular wave of Amplitude 5m and Time period 20 s.



Fig. 7 Roll response of a ship in a regular wave of Amplitude 5m and Time period 40 s.



Fig. 8 Roll response of a ship in a regular wave of Amplitude 5m and Time period 50 s.



Fig. 9 Roll response of a ship in a regular wave of Amplitude 10m and Time period 25 s.



Fig. 10 Roll response of a ship in a regular wave of Amplitude 10m and Time period 50 s.

General observations: It can be observed from the above plots that the roll response increases as the wave amplitude increases and the time period approaches the vessel's natural frequency (52 sec)

8. Irregular beam seas:

Roll to response to irregular seas is modelled in a similar fashion to regular seas. The main difference is the that irregular sea is composed by super imposing waves varying amplitudes and time periods.



Fig.11 Bond graph model of a ship in irregular wave field

Irregular wave equation: The irregular wave is created by superimposing seven different regular waves. Time periods and amplitudes are 1D vectors instead of scalars. The wave generator outputs the irregular waves and wave frequencies to the damping and excitation moment components.

```
parameters
   real T[7] = [10;5;13;12;14;15;26];
   real Amplitude[7] = [1;2;1;0.5;1;2;1];
   real a[7] = [1;1;1;1;1;1;1];

variables
   real Omega[7];
   real Wave[7];

equations
   Omega = 2*pi*((a)./T);
   Wave = (Amplitude).*sin(Omega*time);

   output[1,:] = transpose(Omega);
   output[2,:] = transpose(Wave);

   w = transpose(Omega);
```

Damping: The damping coefficient is calculated independently for each of the seven different waves. The resultant damping coefficient is obtained by considering a parallel connection.

$B_{eq} = \frac{1}{\Sigma \frac{1}{b}}$

```
parameters
   real B = 25.4;
   real T = 9.5;
   real L = 175;
   real global g = 9.81;
   real global density;
   real a[7] = [1;1;1;1;1;1;1];

variables
   real r;
   real b_44[1,7];
   real A;
   real B_44;
   real b;
   real c;
   real d;
   real e[1,7];
   real f[1,7];
   real k[1,7];

equations
   A = B*T;
   b = 0.256*(B/T)-0.286;
   c = -0.11*(B/T)-2.55;
   d = 0.033*(B/T)-1.419;
   k = (w).^(-1.3);
   e = exp(c*k);
   f = (w).^(d);
   b_44 = sqrt(2*g/B)*density*A*B^2*b*(e).*f;
   B_44 = L/(msum(transpose(a)./b_44));
   r = B_44;
   p.e = r * p.f;
```

Excitation moment: Excitation moment is calculated by the adding the moments from each of the component wave.

```
parameters
    real global Displacement;
    real global GM;
    real global g;

variables
    real flow;

    real wave_load[1,7];
    real omega[1,7];
    real wave[1,7];



equations
    omega = input[1,:];
    wave = input[2,:];

    wave_load = Displacement*GM*((omega).^2).*(wave)/g;

    p.e = msum(wave_load);
    flow = p.f;
```

Roll response:

The roll response of the ship in the given irregular wave field is as shown in Fig.12



Fig.12 Roll response of a ship in a irregular seas composed of seven different regular waves

9. References:

[1] Dynamics of Marine Vehicles, Rameshwar Bhattacharya

[2] 20-Sim reference manual, Controllab products

[3] Parametric roll instability of ships, Irfan ahmed Sheik, Master's Thesis, University of Oslo